

ARDUINO - A Tutorial in Connecting and Configuring a Button Switch

(RDB, Stafford (UK),
January 2021)

Contents

Copyright and Warranties.....	3
Audience	4
Introduction	5
Three Considerations	5
Consideration 1 - Type of switch	5
Consideration 2 - Wiring Choices	5
Consideration 3 - Switch Subtleties/ Design/Manufacture Imperfections	7
Putting It All Together – Example Sketch	8
Exercises.....	9
Exercise 1 - Switch Trace	9
Exercise 2 - OOTB.....	9
Exercise 3 – Alternative Switch Circuit, C2.....	10
Exercise 4 – Alternate Switch Read Method.....	10
Exercise 5 – Playing Around With Debounce Values	10
And Finally	11
Example OOTB Sketch S1	12

Copyright and Warranties

This document, associated examples and code is in the public domain and may be used without restriction and without warranty.

Audience

This tutorial has been written to provide a level of basic understanding relating to the connection and reliable use of simple button switches with Arduino microcontrollers. Understanding these fundamentals will ensure that beginners, hobbyists and other users will be able to choose appropriate switch circuit designs and service associated button switches.

The tutorial is aimed at anyone, but particularly beginners and those new to Arduino, needing to 'get under the hood' so to speak. Having said that, beginner or expert user alike, I am sure there will be something for all.

To get the most out of this tutorial the reader should have some basic understanding of circuit wiring (ie using breadboards or otherwise), the Arduino IDE development environment, C/C++ programming language and a few common digital I/O functions native to the Arduino, but specifically, `pinMode()`, `digitalRead()` and `digitalWrite()` (see links, [pinMode](#), [digitalRead](#) [digitalWrite](#)).

If you need to download a copy of the Arduino IDE, see [Arduino - Software](#).

Introduction

In this tutorial we look at simple switches and how these can be connected to Arduino microcontrollers such that they can be used reliably.

The design aims for the tutorial are to:

- be accessible to readers with a basic level of C/C++ programming and awareness of the Arduino architecture – for example, basic digital I/O functions, specifically, `pinMode()`, `digitalRead()`, `digitalWrite()`, Arduino sketch structure, use of the Arduino IDE, etc
- provide a practical appreciation of key considerations in switch implementation, and to underpin these with a real world example by way of a C++ Arduino sketch which will, hopefully, demonstrate the principles laid out.

However, before we crash headlong into switches, wires, resistors, microcontrollers and coding, we will need to consider, appreciate and understand three things:

1. The type of switch to be used
2. How we wish the switch to be wired and configured (there are choices)
3. The fundamental issues inherent with imperfect switch design and manufacture.

Three Considerations

Let's look at each of these considerations in turn:

Consideration 1 - Type of switch

We will look at connecting and configuring single pole push/press switches, or button switches. This is the simplest form of switch – single pole just means it has one input line and one output line which only connect when the switch is pressed/pushed. That is, when activated, the gap between the input and output is closed and so current will flow. However, there are many types of switch and each will have its own characteristics requiring specific software considerations.

Consideration 2 - Wiring Choices

A simple button switch has two terminals each of which will need to be connected to an Arduino pin. One terminal will need to connect to a digital I/O pin used for sensing the operation of the switch when pressed, and the other to either +5v AND 0v (ground) or just 0v (ground), depending on the choice of wiring and, therefore the specific digital sensing required. The wiring scheme of each choice is slightly different. Let's call these two circuits C1 and C2 and consider each in turn.

Figure 1 shows circuit C1 - one terminal connected to a digital I/O pin (the digital read pin), for reading switch change states, and the other terminal to +5v AND 0v (ground) pins on the Arduino, via a 10k ohm resistor.




Figure 1 – Circuit C1

This is the traditional design for wiring a simple switch. The design is such that the digital read pin will initialised as a simple input pin and will be LOW (0v) when the switch is off and HIGH (5v) when pressed (on). The reason for the 10k ohm resistor is that Arduino digital pins are very sensitive to electromagnetic fields that can generate spurious. The addition of a 10k ohm resistor ensures the pin is maintained at 0v unless raised to +5v via operation of the button switch.

However, there is a second and direct way in which a simple switch can be connected to the microcontroller, one that does not suffer from detectable electromagnetic interference.

Figure 2 shows circuit C2 - one terminal connected to a digital pin (the digital read pin), for reading switch change states, and the other terminal to just 0v (ground) pins on the Arduino. Note the absence of a resistor.




Figure 2 – Circuit C2

This approach is less common but does work fine. To note is that switch polarity and sensing is reversed - the design is such that the digital read pin will initialised to be HIGH (+5v) when the switch is off and LOW (0v) when pressed (on).

In either case the digital read pin must be configured, usually in `setup()`, with a `pinMode()` call. Thereafter, testing for switch operation (using `digitalRead()`) will need to take note of the circuit design choice (C1 or C2) as the voltage levels for on and off are a reversal of each other. The following table shows the software parameters for each circuit choice:

Circuit	Setup() <code>pinMode(pin_no, <parameter>)</code>	<code>digitalRead(pin_no)</code> value when switch PRESSED (ON)	<code>digitalRead(pin_no)</code> value when switch RELEASED (OFF)
C1	INPUT	HIGH (+5v) - ON	LOW (0v) - OFF
C2	INPUT_PULLUP	LOW (0v) - ON	HIGH (+5v) - OFF

Table 1 – set up and digital reading

Understanding the set up is critical to be able to reliably configure and operate a switch.

Are we there yet? Nearly there, but not quite. Let's move on to the next consideration – switch subtleties.

Consideration 3 - Switch Subtleties/ Design/Manufacture Imperfections

So far, so good - we have two circuit designs, know how each design is to be initialised and what values we will get when we read a switch, when pressed or not. What more?

Well, the problem with switches, and other mechanical components, is that they do not always operate cleanly. Switches of this simply variety need a little time to change their state and 'settle down'. During this settling time, if we read the value of the switch we will invariably get unreliable results arising from noise generated associated with the physical characteristics and mechanics of operating the switch.

If we used an oscilloscope to look at a typical button switch cycle, going from off to on and back to off, we would see something like the trace shown in figure 3.




Figure 3 - Button Switch OFF/ON/OFF Cycle

The key points to note about this trace are;

1. The figure shows a button switch configured as for circuit C1.
2. The button switch is pressed at time A and released at time C.
3. The total time the button switch is physically pressed is $T = C - A$ seconds.
4. Time T can be as short or as long as the button switch is held down.
5. The state change does not instantaneously go from LOW to HIGH (pressed) or from HIGH to LOW (released) - it takes a finite time for transition to occur, as shown by the time intervals B - A (rising) and D - C (falling).
6. If we read the button switch during these transition times we will get unreliable results, sometimes reading LOW, sometimes reading HIGH.
7. It is not until we read the button switch within the interval C - B that we will get a reliable and consistent reading.

So how do we program our way around these imperfections of the physical world? Well, one common method is to introduce the idea of 'debounce'.

Debounce is a method that accepts the imperfections resulting from noise during state change transitions of switching. The method introduces a timing parameter suitably long to allow a switch to achieve its settled state, either fully and cleanly on, or off. Unless you have an oscilloscope, there is no science to how long this period should be, but it ideally needs to be as short as possible. It is invariably set by trial and error and /or experience. The better the quality of a switch (we would hope) then we would expect the debounce (state transition)

periods to be low, perhaps 10 msecs or less. However, debounce periods of 100 msecs or higher are not uncommon.

There are many examples of how to program switches with debounce on the internet so why have I bothered with the effort to put together this tutorial? Well, many of these examples are not particularly comprehensive. Many simply jump straight into providing some code and little more. My objective has therefore been to provide a fuller appreciation and understanding of the challenges in implementing what is, after all, a fundamental and basic interface – a button switch.

On inspection of the included example sketch, S1, it will be seen that there are two functions designed to read a simple button switch. Whilst both will read the same switch and incorporate debounce logic, they operate in completely different ways. See next section for an appreciation of the differences.

Putting It All Together – Example Sketch

What follows is an example of a sketch, S1, that will reliably read the state of a button switch and which is highly configurable to allow the reader to explore the ideas, concepts, constraints and limitations of using a simple button switch.

The sketch design is such that;

- It includes extensive comments throughout which will hopefully aid the reader in understanding what's going on.
- It supports either choice of switch circuit design (C1 and C2), with minimal change. By default, the sketch is coded to work with circuit C1 ('`circuit_C1`', `INPUT`), but if you wish to use circuit C2 make changes as required, see below.
- The choice of digital input pin used to detect switch state changes is set, by default, to pin 2, but may be changed to any suitable pin as required, see below.
- The onboard LED is used for testing to indicate switch state changes. It is defined using a macro '#define LED `LED_BUILTIN`' so should work okay without installing an external LED/resistor circuit. '`LED_BUILTIN`' is a reserved constant and set by the Arduino IDE depending on the microcontroller board referenced in the IDE configuration data. (see [constants – Arduino Reference](#) – [Defining built-ins: LED_BUILTIN](#).)
- It includes two methods for reading the button switch, each coded as a Boolean function:

Function 1 - `bool read_switch_method_1()`

This version of switch reading examines switch once each time the function is called. It allows the code section from which it is called to continue without waiting for the switch press cycle to complete once switching is initiated. The only drawback, of course, is that the design of the calling code must ensure that the switch is regularly tested to catch a change in switch status. However, this should not normally be an issue or concern.

Function 2 - `bool read_switch_method_2()`

This version of switch reading will wait for a switch cycle to complete once it is initiated, before control is returned to the calling code. That is, the calling code will be held up once the switch is pressed and until the debounce period has elapsed. Once the switch is pressed the function keeps total processing control for a time equivalent to at least the debounce period. This version is therefore less efficient than `read_switch_method_1()`.

- The control code within the main sketch loop may incorporate either `read_switch_method_1()` or `read_switch_method_2()`. By default,

`read_switch_method_1()` is coded in the main loop call, it being more efficient. This can be changed if desired, see below.

In summary, out of the box (OOTB) the sketch (S1) is configured as follows:

Configurable Parameters	Sketch Variable/Definitions	Sketch Default Values
Switch circuit design	<code>#define circuit_type</code>	<code>circuit_C1</code>
Digital input pin for switch reading	<code>int button_switch =</code>	<code>2</code>
LED output pin	<code>#define LED</code>	<code>LED_BUILTIN</code>
Debounce period (msecs)	<code>#define debounce</code>	<code>50</code>
Main loop switch read function call		<code>read_switch_method_1()</code>

Table 2 – Out Of The Box Configuration


Exercises

The following exercises are suggested to help consolidate and explore understanding.

Exercise 1 - Switch Trace

Figure 3 shows an example of an oscilloscope trace for a button switch connected as for circuit C1 (`pinMode(..., INPUT)`).

Draw a trace for a button switch designed with circuit C2 (`pinMode(..., INPUT_PULLUP)`).



Exercise 2 - OOTB

Load the OOTB sketch and connect a button switch in the configuration of circuit C1 to your microcontroller. That is, with a pull down resistor, see Figure 1.

The sketch definition and variable values should be as in Table 2, above, but if not make the changes to reflect those in Table 2.

Compile and up load the sketch and operate the button switch in your circuit. In particular, note that the LED toggles on/off with each switch press and that the switch state change only occurs when it is released.

Exercise 3 – Alternative Switch Circuit, C2

Reconfigure the button switch design to now follow circuit C2, see Figure 2. Edit the macro definition `#define circuit_type circuit_C1` to now be `#define circuit_type circuit_C2`.

The sketch definitions and variable value should be as below:

Configurable Parameters	Sketch Variable/Definitions	Sketch Default Values
Switch circuit design	<code>#define circuit_type</code>	<code>circuit_C2</code>
Digital input pin for switch reading	<code>int button_switch =</code>	2
LED output pin	<code>#define LED</code>	<code>LED_BUILTIN</code>
Debounce period (msecs)	<code>#define debounce</code>	50
Main loop switch read function call		<code>read_switch_method_1()</code>

Compile and up load the sketch. What do you notice? The sketch should operate exactly the same as for the OOTB settings with `circuit_C1`.

Exercise 4 – Alternate Switch Read Method

Now use the alternate switch read function by editing `read_switch_method_1()` in the main code loop to be `read_switch_method_2()`.

The sketch definitions and variable value should be as below:

Configurable Parameters	Sketch Variable/Definitions	Sketch Default Values
Switch circuit design	<code>#define circuit_type</code>	<code>circuit_C2</code>
Digital input pin for switch reading	<code>int button_switch =</code>	2
LED output pin	<code>#define LED</code>	<code>LED_BUILTIN</code>
Debounce period (msecs)	<code>#define debounce</code>	50
Main loop switch read function call		<code>read_switch_method_2()</code>

What do you notice? The sketch operates exactly the same as for the exercises.

Exercise 5 – Playing Around With Debounce Values

Using the switch circuit and sketch as for exercise 4, vary the debounce value (`#define debounce`) and see how decreasing/increasing this affects the switch's operation. Note the results for your future use.

The sketch definitions and variable value should be as below:

Configurable Parameters	Sketch Variable/Definitions	Sketch Default Values
Switch circuit design	<code>#define circuit_type</code>	<code>circuit_C1</code>
Digital input pin for switch reading	<code>int button_switch =</code>	2
LED output pin	<code>#define LED</code>	<code>LED_BUILTIN</code>
Debounce period (msecs)	<code>#define debounce</code>	0, 25, 100, 150, 200, 250, etc.
Main loop switch read function call		<code>read_switch_method_1()</code>

And Finally

I hope that, through this tutorial, the reader has reached a fuller appreciation and understanding of connecting and using simple switches with Arduino microcontrollers.

The simple button switch is commonly used but there are many choices and examples of switch. Whilst the majority are of a digital nature (i.e. either off or on), we should not dismiss analogue components that can also be used as inputs for switching, e.g. potentiometers, photoresistors, etc. Each will invariably have its own peculiarities which will need to be understood and suitably programmed for. The good news is that the coding does not get necessarily more complex.

Enjoy!

Example OOTB Sketch S1

```
// This example and code is in the public domain and may be used without restriction and
// without warranty.

/*
  READING SIMPLE SWITCHES RELIABLY, WITH OR WITHOUT A SWITCH PULL DOWN RESISTER
  .....
```

In this example sketch we look at configuring a simple button switch such that when pressed it will toggle a LED on and off. The sketch automatically allows for one of two switch circuits to be configured, either with a pull down switch resister or without. This is configured and controlled with a simple macro parameter - see points of note below.

Additionally, this sketch offers two methods to read a simple button switch, each offered as a specific function. Only one function method should be configured (..._method_1/..._method_2), the choice left to preference of the user. The differences between each method is highlighted and explained below:

Method 1 - Function 'bool read_switch_method_1()' - this version of switch reading examines the switch input once each time the function is called. This allows the code section from which it is called to continue without waiting for the switch press cycle to complete once switching is initiated. Only drawback, of course, is that the design of the calling code must ensure that the switch is regularly tested to catch a change in switch status.

Method 2 - Function 'bool read_switch_method_2()' - this version of switch reading will wait for a switch cycle to complete, once it is initiated, before control is returned to the calling code. Once the switch is pressed, the code will fully consume the debounce period AFTER switch release until control is returned back to the calling code. That is, the calling code will be held up once the switch is pressed and until the debounce period has elapsed.


Points of note:

1. The digital pin chosen as the input is initialised according to the 'circuit_type' macro parameter, either INPUT_PULLUP or INPUT using a call to pinMode(button_switch,circuit_type) in the setup() function. As the conditions for detecting switch on/off are different for each 'circuit type' (they are reversed) two variables are used ('switch_low' and 'switch_high') to provide a reference


indicating low and high (on/off) conditions. However, this is transparent to the functioning of the code. The only requirement is to define the 'circuit_type', everything else is automatically taken care of.

2. The design for each reading method incorporates code to debounce spurious inputs when the switch is pressed which, if not accounted for, would produce unexpected/spurious results.
3. The wiring designs for each type of switch circuit are:

With switch pull down resister configured and 'circuit_type' of INPUT



With NO switch pull down resister configured and 'circuit_type' of INPUT_PULLUP



5. The on board LED (MEGA 2560) is utilised for testing to keep circuit design to a minimum. If not using the onboard LED then configure one in the traditional way using a 230ohm resister (for a red LED) and a red LED on a breadboard, or otherwise.

*/

```
74
75 // define type of switch circuit and associated LOW and HIGH variables:
76 //   - circuit C1, INPUT with switch resistor      -> switch_high = HIGH, switch_low = LOW
77 //   - circuit C2, INPUT_PULLUP with no switch resistor -> switch_high = LOW, switch_low = HIGH
78
79 #define circuit_C1      INPUT
80 #define circuit_C2      INPUT_PULLUP
81
82 #define circuit_type    circuit_C1    // circuit type configured, see circuit design specs
83
84 int      switch_high, switch_low;
85
86 #define button_switch    2            // digital pin connected to button switch
87 #define debounce        50           // number of milliseconds to wait for switch to settle once pressed
88 #define switched        true         // signifies switch has been pressed
89
90 #define LED              LED_BUILTIN  // digital pin connected to LED, for testing of switch code only
91 bool    led_status      = LOW;       // start with LED off, for testing of switch code only
92
```

```
93
94 void setup() {
95     // define the switch circuit type
96     pinMode(button_switch, circuit_type); // circuit_type == INPUT or INPUT_PULLUP
97     // establish meanings for switch on/off depending on circuit_type
98     if (circuit_type == INPUT_PULLUP) {
99         // switch is NOT configured with a pull down switch resistor
100         switch_high = LOW; // switch pin goes LOW when switch pressed, ie on
101         switch_low = HIGH; // switch pin goes HIGH when switch released, ie off
102     } else {
103         // circuit_type == INPUT, so switch IS configured with a pull down switch resistor
104         switch_high = HIGH; // switch pin goes HIGH when switch pressed, ie on
105         switch_low = LOW; // switch pin goes LOW when switch released, ie off
106     }
107     // set LED pin for output, for testing purposes only
108     pinMode(LED, OUTPUT);
109 }
110
```

```

111 //
112 // Button switch reading, method 1
113 // This version of switch reading examines switch once each time the function is called.
114 // This allows the code section from which it is called to continue without waiting
115 // for the switch press cycle to complete once switching is initiated.
116 // Only drawback, of course, is that the design of the calling code must ensure that the
117 // switch is regularly tested to catch a change in switch status.
118 //
119 //
120 //
121 bool read_switch_method_1() {
122     int          switch_pin_reading;
123     // static variables because we need to retain old values between function calls
124     static bool   switch_pending = false;
125     static long int elapse_timer;
126     switch_pin_reading = digitalRead(button_switch);
127     if (switch_pin_reading == switch_high) {
128         // switch is pressed, so start/restart debounce process
129         switch_pending = true;
130         elapse_timer = millis();    // start elapse timing
131         return !switched;          // now waiting for debounce to conclude
132     }
133     if (switch_pending && switch_pin_reading == switch_low) {
134         // switch was pressed, now released, so check if debounce time elapsed
135         if (millis() - elapse_timer > debounce) {
136             // dounce time elapsed, so switch press cycle complete
137             switch_pending = false;
138             return switched;
139         }
140     }
141     return !switched;
142 }
143

```



```
144
145 //
146 // Button switch reading, method 2.
147 // This version of switch reading will wait for a switch cycle to complete once it
148 // initiated, before control is returned to the calling code.
149 // That is, the calling code will be held up once the switch is pressed and until
150 // the debounce period has elapsed.
151 //
152 bool read_switch_method_2() {
153     int      switch_pin_reading;
154     bool      switch_status;
155     long int  elapse_timer;
156     switch_status = !switched; // assume switch not pressed
157     // read the given switch_pin, if pressed will be HIGH, if not pressed will be LOW
158     switch_pin_reading = digitalRead(button_switch);
159     do {
160         if (switch_pin_reading == switch_high) {
161             // switch has been pressed, now debounce
162             switch_status = switched; // flag that switch was pressed
163             elapse_timer = millis();
164             do {} while (millis() - elapse_timer < debounce); // wait for debounce period to lapse
165             switch_pin_reading = digitalRead(button_switch); // see if switch is still being pressed
166         }
167     } while (switch_pin_reading == switch_high); // keep debouncing until switch no longer pressed
168     return switch_status; // result is either 'switched' or '!switched'
169 }
170
```

```
171
172 void loop() {
173     do {
174         if (read_switch_method_1() == switched) {
175             led_status = HIGH - led_status;           // flip between HIGH and LOW
176             digitalWrite(LED, led_status);
177         }
178     } while (true);
179 }
```